

# Alloy

## Specifikace pomocí relační logiky

Radek Mařík

Czech Technical University  
Faculty of Electrical Engineering  
Department of Telecommunication Engineering  
Prague CZ

November 9, 2019



## 1 Alloy

- Motivace
- Základní prvky jazyka
- Nástroj Alloy

## 2 Studijní příklady

- Stropy a podlahy
- Vybranné konstrukce
- Dynamické systémy

# Outline

## 1 Alloy

- Motivace
- Základní prvky jazyka
- Nástroj Alloy

## 2 Studijní příklady

- Stropy a podlahy
- Vybranné konstrukce
- Dynamické systémy

# I Am My Own Grandpa - píseň

- Výzvou je vytvořit situaci muže, který je svým vlastním dědečkem, aniž by byl spáchán incest nebo bylo potřeba cestování časem.
- Slova písně popisují řešení.

# I Am My Own Grandpa

Many many years ago, when I was twenty-three,  
I was married to a widow as pretty as can be,  
This widow had a grown-up daughter who had hair of red,  
My father fell in love with her and soon the two were wed.

I'm my own grandpa, I'm my own grandpa.  
It sounds funny, I know, but it really is so  
I'm my own grandpa.

This made my dad my son-in-law and changed my very life,  
For my daughter was my mother, for she was my father's wife.  
To complicate the matter, even though it brought me joy,  
I soon became the father of a bouncing baby boy.

My little baby thus became a brother-in-law to dad,  
And so became my uncle, though it made me very sad,  
For if he was my uncle then that also made him brother  
To the widow's grown-up daughter, who of course was my step-mother.



# ... by Dwight B. Latham and Moe Jaffe

Father's wife then had a son who kept them on the run.  
And he became my grandchild for he was my daughter's son.  
My wife is now my mother's mother and it makes me blue,  
Because although she is my wife, she's my grandmother, too.

Oh, if my wife's my grandmother then I am her grandchild.  
And every time I think of it, it nearly drives me wild.  
For now I have become the strangest case you ever sawn  
As the husband of my grandmother, I am my own grandpa.

I'm my own grandpa, I'm my own grandpa.  
It sounds funny, I know, but it really is so I'm my own grandpa.  
I'm my own grandpa, I'm my own grandpa.  
It sounds funny, I know, but it really is so  
I'm my own grandpa.



# I Am My Own Grandpa - Alloy řešení

```

module grandpa
abstract sig Person {
  father: lone Man,
  mother: lone Woman }
sig Man extends Person { wife: lone Woman }
sig Woman extends Person { husband: lone Man }
fact Biology { no p: Person | p in p.^(mother+father) }
fact Terminology { wife = ~husband }
fact SocialConvention {
  no wife & *(mother+father).mother
  no husband & *(mother+father).father }
fun grandpas [p: Person]: set Person {
  let parent = mother + father + father.wife + mother.husband |
  p.parent.parent & Man }
pred ownGrandpa [m: Man] { m in grandpas[m] }
run ownGrandpa for 4 Person expect 1

```



# Alloy - pro co se používá?

- Alloy je modelovací jazyk pro návrh softwaru.
- *??? Není však určen pro modelování architektury (jako např. UML).*
- Je dostatečně obecný, aby mohl modelovat
  - jakoukoliv doménu individuálů,
  - relace mezi nimi.
- **syntaxe Alloy 4.2 - 5.1** (pozor na tutorialy pro 3.x)
- **Styl použití**
  - 1 Specifikace podmínek modelu
    - struktury a relace mezi nimi
    - podmínky jako obecně fakta
  - 2 Hledání řešení či protipříkladu
    - Řešení: hledání instance modelu, která splňuje všechny podmínky
    - Verifikace: konstrukce domněnky (EN conjecture)
      - buď nelze vyvrátit
      - nebo se nalezne protipříklad





# Outline

- 1 Alloy
  - Motivace
  - Základní prvky jazyka
  - Nástroj Alloy
- 2 Studijní příklady
  - Stropy a podlahy
  - Vybranné konstrukce
  - Dynamické systémy

# Atomy

- Vše je postaveno na atomech a relacích.
- **Atom** je primitivní entita, která je
  - **nedělitelná**: nemůže být rozdělena na menší části,
  - **neměnná**: její vlastnosti se nemění v čase,
  - **neinterpretovaná**: nemá žádnou vestavěnou vlastnost,
- **Relace** je struktura, která zachycuje vztahy mezi atomy.  
Je to množina  $n$ -**tic**, každá  $n$ -tice je sekvence atomů




# Signatura

- **Signatura** zavádí množinu atomů.
- Deklarace definující množinu pojmenovanou  $A$   
`sig A { }`
- Množina může být zavedena jako **rozšíření** jiné množiny, takto  $A1$  je podmnožinou množiny  $A$ .  
`sig A1 extends A { }`
- Signatury deklarované nezávisle na jakékoliv ostatní signatuře je tzv. **vrcholová** signatura.
- Rozšíření té samé signatury jsou vzájemně disjunktní, tak jako vrcholové signatury.
- Množina může být zavedena jako **podmnožina** jiné množiny  
`sig A1 in A { }`
- **Abstraktní signatura** nemá žádné elementy s výjimkou těch, které patří do jejích rozšíření nebo podmnožin.

`abstract sig A { }`

# Signatura - příklad

```
abstract sig Person { }  
sig Man extends Person { }  
sig Woman extends Person { }  
sig Married in Person { }
```



# Pole

- Relace se deklarují jako **pole** v signatuře.
- Deklarace definující relaci  $f$ ,
  - jejíž doména je  $A$  a
  - jejíž obraz je dán výrazem  $e$

`sig A { f: e }`

## Příklady

- Binární relace ...  $f1$  je podmnožinou  $A \times A$

`sig A { f1: A }`

- Ternární relace ...  $f2$  je podmnožinou  $B \times A \times A$

`sig B { f2: A -> A }`



# Pole

- Relace se deklarují jako **pole** v signatuře.
- Deklarace definující relaci  $f$ ,
  - jejíž doména je  $A$  a
  - jejíž obraz je dán výrazem  $e$

`sig A { f: e }`

## Příklady

- Binární relace ...  $f1$  je podmnožinou  $A \times A$

`sig A { f1: A }`

- Ternární relace ...  $f2$  je podmnožinou  $B \times A \times A$

`sig B { f2: A -> A }`



# Násobnost

- Dovoluje nám omezit velikost množin.
  - Klíčové slovo násobnosti umístěné před deklarací signaturou omezuje počet elementu v množině signatury

$$m \text{ sig } A \{ f: e \}$$

- Můžeme omezit násobnosti polí

$$\text{sig } A \{ f: m e \}$$

$$\text{sig } A \{ f: e1 \ m \rightarrow \ n \ e2 \}$$

- Existují čtyři násobnosti
  - **set**: jakýkoliv počet,
  - **some**: jeden nebo více,
  - **lone**: nula nebo jeden (L),
  - **one**: přesně jeden,
- Implicitní klíčové slovo, pokud je vynecháno, je **one**. Takže následující jsou ekvivalentní zápisy:

$$\text{sig } A \{ f: e \}$$

$$\text{sig } A \{ f: \text{one } e \}$$


## Relace - příklad 2

```
abstract sig Person {  
  children: set Person,  
  siblings: set Person,  
}  
sig Man, Woman extends Person { }  
sig Married in Person {  
  spouse: one Married,  
}
```





# Kvantifikátory

- Alloy poskytuje bohatou kolekci kvantifikátorů
  - `all`  $x: S \mid F$  :  $F$  platí pro každé  $x$  v  $S$ ,
  - `some`  $x: S \mid F$  :  $F$  platí pro nějaké  $x$  v  $S$ ,
  - `no`  $x: S \mid F$  :  $F$  selže pro každé  $x$  v  $S$ ,
  - `lone`  $x: S \mid F$  :  $F$  platí pro nejvíce jedno  $x$  v  $S$ ,
  - `one`  $x: S \mid F$  :  $F$  platí pro právě jedno  $x$  v  $S$ ,



# Logické operátory

- Jsou k dispozici běžně používané logické operátory

<code>not</code>	<code>!</code>	negace
<code>and</code>	<code>&amp;&amp;</code>	konjunkce
<code>or</code>	<code>  </code>	disjunkce
<code>implies</code>	<code>=&gt;</code>	implikace
<code>else</code>		alternativa
	<code>&lt;=&gt;</code>	iff (ekvivalence)

- Příklad

`a != b` je ekvivalentní `not a = b`

# Množiny a operátory

- Předdefinované množinové konstanty

`none` : prázdná množina,

`univ` : univerzální množina,

`ident` : identita,

- Množinové operátory

`+` : sjednocení

`&` : průnik

`-` : rozdíl

`in` : podmnožina

`=` : rovnost

- Příklad: ženatí muži

`Married & Man`

- Vymezená množina (angl. set comprehension)

- Množina hodnot množiny  $S$ , pro které platí  $F$

$\{ x : S \mid F \}$

# Relační operátory

- > šipka (součin)
- ~ transpozice
- . dot join
- box join
- ^ tranzitivní uzávěr
- \* reflexivně-tranzitivní uzávěr
- <: omezení domény
- :> omezení obrazu
- ++ přepsání

# Součin, Transpozice

- Šipkový součin  $p \rightarrow q$

- $p$  a  $q$  jsou dvě relace
- $p \rightarrow q$  je relace, která vezme všechny kombinace  $n$ -tic relace  $p$  a  $m$ -tic relace  $q$  a spojí je (konkatenace).
- Příklad

Name = { (N0), (N1) }

Addr = { (D0), (D1) }

Book = { (B0) }

Book  $\rightarrow$  Name  $\rightarrow$  Addr = { (B0,N0,D0), (B0,N0,D1),  
(B0,N1,D0), (B0,N1,D1) }

- Transpozice  $\sim p$

- produkuje zrcadlový obraz relace  $p$
- tzn. reverzuje pořadí atomů v každé  $n$ -tici.
- Příklad

example = { (a0,a1,a2,a3), (b0,b1,b2,b3) }

$\sim$ example = { (a3,a2,a1,a0), (b3,b2,b1,b0) }



# Spojení n-tic

- **p.q** Co je spojením těchto dvou n-tic?
  - $p = (s_1, \dots, s_n)$
  - $q = (t_1, \dots, t_m)$
  - Jestliže  $s_n \neq t_1$ , potom výsledek je prázdný
  - Jestliže  $s_n = t_1$ , potom výsledkem je k-tice  $(s_1, \dots, s_{n-1}, t_2, \dots, t_m)$
- Příklad
$$\{(a, b)\} \cdot \{(a, c)\} = \{\}$$
$$\{(a, b)\} \cdot \{(b, c)\} = \{(a, c)\}$$
- Co se stane v případě  $\{(a)\} \cdot \{(a)\}$ ?
  - Není definováno!
  - $p.s$  je definováno, tedy a jen tehdy, jestliže  $p$  a  $s$  nejsou obě unární relace



# Spojení relací, Uzávěry, Omezení relací

- $p \cdot q$ 
  - $p$  a  $q$  jsou dvě relace a nejsou obě unární.
  - $p \cdot q$  je relace, která vznikne kombinací všech  $n$ -tic z  $p$  a  $m$ -tic z  $q$  a přidáním jejich spojení, pokud tento existuje.
- $p[q]$  (box join)
  - sémanticky identické spojení dot join, ale bere své argumenty v opačném pořadí

$$p[q] \equiv q \cdot p$$

- $\hat{r} = r + r \cdot r + r \cdot r \cdot r + \dots$
- $*r = \hat{r} + \text{iden}$
- $s <: r$  obsahuje  $n$ -tice z  $r$  **začínající** elementem v  $s$   
( $\text{range}(s <: r) = s \cdot r$ )
- $r >: s$  obsahuje  $n$ -tice z  $r$  **končící** elementem v  $s$   
( $\text{domain}(r >: s) = r \cdot s$ )
- $p ++ q = p - (\text{domain}(q) <: p) + q$



## Příklad

```

module grandpa
abstract sig Person {
  father: lone Man,
  mother: lone Woman }
sig Man extends Person { wife: lone Woman }
sig Woman extends Person { husband: lone Man }
fact Biology { no p: Person | p in p.^(mother+father) }
Man = {(Jirka), (Tomas), (Josef), (Vlada), (Franc)}
Woman = {(Jana), (Lenka), (Tereza), (Olga)}
father = {(Jirka, Tomas), (Lenka, Tomas), (Tomas, Josef),
          (Josef, Vlada), (Jana, Franc)}
mother = {(Jirka, Jana), (Jana, Tereza), (Tomas, Olga)}
{(Jirka)}.father = {(Tomas)}
{(Jirka)}.mother = {(Jana)}
{(Jirka)}.father.father = {(Josef)}
{(Jirka)}.father.mother = {(Olga)}
{(Jirka)}.mother.father = {(Franc)}
{(Jirka)}.mother.mother = {(Tereza)}

```





## Příklad

```

module grandpa
abstract sig Person {
  father: lone Man,
  mother: lone Woman }
sig Man extends Person { wife: lone Woman }
sig Woman extends Person { husband: lone Man }
fact Biology { no p: Person | p in p.^(mother+father) }
Man = {(Jirka), (Tomas), (Josef), (Vlada), (Franc)}
Woman = {(Jana), (Lenka), (Tereza), (Olga)}
father = {(Jirka,Tomas), (Lenka,Tomas), (Tomas,Josef),
  (Josef, Vlada), (Jana, Franc)}
mother = {(Jirka, Jana), (Jana, Tereza), (Tomas, Olga)}
father + mother = {(Jirka,Tomas), (Lenka,Tomas), (Tomas,Josef),
  (Josef, Vlada), (Jana, Franc), (Jirka, Jana),
  (Jana, Tereza), (Tomas, Olga)}
{(Jirka)}.(father+mother) = {(Tomas), (Jana)}
{(Jirka)}.(father+mother).(father+mother) =
  {(Josef), (Olga), (Franc), (Tereza)}
{(Jirka)}.(father+mother).(father+mother).(father+mother) = {(Vlada)}

```



# Let

- Lze zjednodušit výrazy

```
let x = e | A
```

- Každý výskyt proměnné  $x$  je nahrazen výrazem  $e$  v  $A$

- Příklady

- "Člověk v manželství má právě jednoho partnera"

```
sig Married in Person { spouse: one Married }
```

- "Každý ženatý muž (žena) má manželku (manžela)"

```
all p: Married |  
  let q = p.spouse |  
    (p in Man => q in Woman) and  
    (p in Woman => q in Man)
```



# Skaláry

- Vše je množina v Alloy

- neexistují skaláry
- používáme singleton relaci

```
let matt = one Person
```

- Interpretace kvantifikací:

```
all x : S | ... x ...
```

$x = \{t\}$  pro nějaký element  $t$  z  $S$

# Fakta

- Další omezení na signatury a pole lze vyjádřit v Alloy jako **fakta**
- AA (Alloy Analyzer) hledá instance modelu, které také splňují všechna omezení určená fakty
- Příklad: "Žádná osoba nemůže být svým vlastním předchůdcem."

```
fact selfAncestor {  
  no p: Person | p in p.^parents  
}
```



# Funkce a predikáty

- Mohou být použity jako "makra"
  - Mohou být pojmenované a násobně použité v různých kontextech (fakta, tvrzení, podmínky běhu)
  - Mohou být parametrizována, používají se ke zkrácení zápisů
- **Funkce:**
  - Pojmenovaný výraz s žádným či více argumenty.
  - Vrací výraz jako návratovou hodnotu
  - Během analýzy se vyvolávají pouze, pokud je na ně odkázáno jménem.
  - Příklad: "Relace rodiče."

```
fun parents []: Person -> Person { ~children }
```

- Příklad: "Sestry."

```
fun sisters [p: Person]:
  { {w: Woman | w in p.siblings } }
```

- Příklad: "Žádný člověk nemůže být svým vlastním předkem nebo sestrou."

```
all p: Person |
  not (p in p.^parents or p in sisters[p])
```



# Predikáty

**Predikáty** jsou dobré v situacích:

- Omezení, která nechcete zaznamenat jako fakta.
- Omezení, která chcete použít vícekrát
- Během analýzy se vyvolávají pouze, pokud je na ně odkázáno jménem.
- Příklad: "Dvě osoby jsou pokrevní příbuzní, jestliže mají společného předka."

```
pred BloodRelated [p: Person, q: Person] {  
    some p.*parents & q.*parents  
}
```

- Příklad: "Osoba nemůže být v manželství s pokrevním příbuzným."  
`no p: Married | BloodRelated [p, p.spouse]`



# Outline

## 1 Alloy

- Motivace
- Základní prvky jazyka
- **Nástroj Alloy**

## 2 Studijní příklady

- Stropy a podlahy
- Vybranné konstrukce
- Dynamické systémy

# Příkaz Run

- **run** příkaz způsobí, že AA jeho provedením analyzuje model.
- Příkazuje nástroji, aby hledal instance modelu.
- AA provede pouze první příkaz **run** v souboru.
- **AA hledá pouze v omezeném prostoru instancí určeném rozsahem.**
- **Rozsah** (angl. scope) reprezentuje maximální počet n-tic v každé vrcholové signatuře.
- Přednastavená hodnota rozsahu = 3
- Příklady

```
run {} /* the scope is 3 */  
run { for 5 /* the scope is 5 */  
run { some Man && no Married } /* with conditions */
```





# Tvrzení

- Často věříme, že náš model splňuje jisté omezení, vlastnost, která není přímo vyjádřena.
- Můžeme nadefinovat taková dodatečná omezení jako tvrzení a použít AA k jejich ověření.
- Pokud omezení vyjádřené daným tvrzením není splněno, AA vyprodukuje instanci protipříkladu.
- Příklady
  - "Žádná osoba nemá rodiče, který je zároveň bratrancem či sestřenicí."

```
assert a1 {all p: Person |
  no p.parents & p.siblings }
```



## I Am My Own Grandpa - Alloy řešení

```

module grandpa
abstract sig Person {
  father: lone Man,
  mother: lone Woman }
sig Man extends Person { wife: lone Woman }
sig Woman extends Person { husband: lone Man }
fact Biology { no p: Person | p in p.^(mother+father) }
fact Terminology { wife = ~husband }
fact SocialConvention {
  no wife & *(mother+father).mother
  no husband & *(mother+father).father }
fun grandpas [p: Person]: set Person {
  let parent = mother + father + father.wife + mother.husband |
  p.parent.parent & Man }
pred ownGrandpa [m: Man] { m in grandpas[m] }
run ownGrandpa for 4 Person expect 1

```



# Outline

- 1 Alloy
  - Motivace
  - Základní prvky jazyka
  - Nástroj Alloy
- 2 Studijní příklady
  - Stropy a podlahy
  - Vybranné konstrukce
  - Dynamické systémy

# Úloha stropů a podlah - zadání

Písnička Paula Simona, 1973

*“One Man’s Ceiling Is Another Man’s Floor”*

```
module CeilingsAndFloors
  sig Platform {}
  sig Man {ceiling, floor: Platform}
  fact PaulSimon {all m: Man | some n: Man | n.Above[m]}
  pred Above[m, n: Man] {m.floor = n.ceiling}
```




# Úloha stropů a podlah - protipříklad řešení

```
open CeilingsAndFloors
```

```
assert BelowToo { all m: Man | some n: Man | m.Above[n] }
```

```
check BelowToo for 2 expect 1
```




# Úloha stropů a podlah - protipříklad řešení

```

open CeilingsAndFloors
assert BelowToo { all m: Man | some n: Man | m.Above[n] }
check BelowToo for 2 expect 1

```



John Mc Naughton's Twisted House

# Úloha stropů a podlah - protipříklad řešení s geometrií

```

open CeilingsAndFloors
pred Geometry {no m: Man | m.floor = m.ceiling}
assert BelowToo' { Geometry =>
  (all m: Man | some n: Man | m.Above[n]) }
check BelowToo' for 2 expect 0

```


Executing "Check BelowToo' for 2 expect 0"

Solver=sat4j Bitwidth=0 MaxSeq=0 SkolemDepth=1 Symmetry=20  
 211 vars. 18 primary vars. 335 clauses. 15ms.  
 No counterexample found. Assertion may be valid, as expected. 16ms.

```

check BelowToo' for 3 expect 1

```



# Úloha stropů a podlah - řešení s NoSharing

```
open CeilingsAndFloors
```

```
pred NoSharing {  
  no m,n: Man | m!=n  
  && (m.floor = n.floor || m.ceiling = n.ceiling) }
```

```
assert BelowToo'' { NoSharing  
  => (all m: Man | some n: Man | m.Above[n]) }  
check BelowToo'' for 6 expect 0  
check BelowToo'' for 10 expect 0
```

Executing "Check BelowToo'' for 10 expect 0"

Solver=sat4j Bitwidth=0 MaxSeq=0 SkolemDepth=1 Symmetry=20

6750 vars. 330 primary vars. 14472 clauses. 296ms.

No counterexample found. Assertion may be valid, as expected. 889ms.





# Outline

- 1 Alloy
  - Motivace
  - Základní prvky jazyka
  - Nástroj Alloy
- 2 Studijní příklady
  - Stropy a podlahy
  - **Vybranné konstrukce**
  - Dynamické systémy

# Blue Planet Puzzles

- Blue planet creatures:
  - insane geniuses (blue creatures), who sometimes tell the truth and sometimes lie,
  - normals (yellow), who always lie, and
  - sane geniuses (green), who always tell the truth.
- The Master of Universe poses puzzles so that in each of the puzzles there is always only one insane genius, one sane genius and one normal.
- **Problem 1**
  - A: I am blue.
  - B: I am green if A is a normal.
  - C: I am yellow and A's statement is true.
  - *Who is what?*



# Bijektivní zobrazení - vzájemně jednoznačné zobrazení I

```
module SaneInsaneNormalOnBluePlanet
open util/boolean // Bool, True, False

abstract sig Color {}
one sig Blue, Green, Yellow extends Color {}

abstract sig Speaker {}
one sig Sane, Insane, Normal extends Speaker {}

abstract sig Creature { // the relation table
  speaker: Speaker,
  color: Color,
  speech: Bool,
}
```



# Bijektivní zobrazení - (1:1) omezení, II

```
pred IsSane [a:Creature]
  {a.speaker = Sane and a.speech = True}
pred IsNormal [a:Creature]
  {a.speaker = Normal and a.speech = False}
pred IsInsane [a:Creature]
  {a.speaker = Insane}
pred TrueSentence [a: Creature]
  {IsSane[a] or IsNormal[a] or IsInsane[a]}

fact BluePlanet {
  all c: Creature | TrueSentence[c]
  Blue.~color.speaker = Insane // (1:1) correspondences
  Yellow.~color.speaker = Normal
  Green.~color.speaker = Sane
}
```



# Bijektivní zobrazení - zadání, III

```

fact MasterPuzzles { // (1:1) mappings
  all s: Speaker | one s.~speaker
  all c: Color | one c.~color
}
one sig A, B, C extends Creature {}

pred Problem1 [] {
  A.speech=True <=> A.color = Blue
  B.speech=True <=> (A.speaker = Normal => B.color = Green)
  C.speech=True <=> (A.speech = True and C.color = Yellow)
}
run {Problem1}

```

**Řešení:** A is a normal, B a sane genius and C an insane genius.



# Uspořádání - řada domů

```

open util/ordering[House] as hsOrd
abstract sig House {}
one sig H1, H2, H3, H4 extends House {}{
  hsOrd/next[H1] = H2
  hsOrd/next[H2] = H3
  hsOrd/next[H3] = H4 }
abstract sig Nationality {
  lives: one House,
  drinks: one Drink,
  owns: one Animal,
  smokes: one Cigarette, }
fact {
  H3.~lives.drinks = Milk //9
  Chesterfields.~smokes.lives in
  ( hsOrd/prev[Fox.~owns.lives]
    + hsOrd/next[Fox.~owns.lives]) //11 }

```



# Outline

- 1 Alloy
  - Motivace
  - Základní prvky jazyka
  - Nástroj Alloy
- 2 Studijní příklady
  - Stropy a podlahy
  - Vybranné konstrukce
  - **Dynamické systémy**

# Dynamické modely

- modelování systému se **stavy** a **přechody**
- modelování **operací**, které způsobují přechody

## Alloy

- nezná koncept stavového přechodu,
- existuje několi způsobů, jak dynamiku systému modelovat
  - specifikace signatury Time vyjadřující čas
  - přidat časovou komponentu do každé relace, která se mění s časem.



# Dynamické modely

- modelování systému se **stavy** a **přechody**
- modelování **operací**, které způsobují přechody

## Alloy

- nezná koncept stavového přechodu,
- existuje několi způsobů, jak dynamiku systému modelovat
  - specifikace signatury Time vyjadřující čas
  - přidat časovou komponentu do každé relace, která se mění s časem.

# Statický model rodiny

```
abstract sig Person {  
  children: set Person,  
  siblings: set Person,  
}
```

```
sig Man, Woman extends Person { }
```

```
sig Married in Person {  
  spouse: one Married,  
}
```



# Dynamický model rodiny

```
sig Time {}
```

```
abstract sig Person {  
  children: Person set -> Time,  
  siblings: Person set -> Time,  
}
```

```
sig Man, Woman extends Person { }
```

```
sig Married in Person {  
  spouse: Married one -> Time,  
}
```

Signatury jsou časově nezávislé

- Married není správně modelovaná

# Vyjádření přechodu v Alloy I

- Přechod může být modelován jako predikát mezi dvěma stavy:
  - **stav těsně před** přechodem a
  - **stav právě po** přechodu.
- Definováno jako predikát s (nejméně) dvěma formálními parametry:  
 $t, t' : \text{Time}$
- Omezení, která vymezují stav v obou časech.



# Vyjádření přechodu v Alloy II

- **Vstupní podmínky**
  - Co platí ve stavech před přechodem.
- **Výstupní podmínky**
  - Popis efektů přechodů, které generují následující stav
- **Invariant**
  - Popis toho, co se nemění
- **Doporučuje se dobře komentovat jednotlivé kategorie podmínek a invariantu**



# Protokol klíčů v hotelu - základní signatury

```
module hotel
open util/ordering [Time] as TO
open util/ordering [Key] as KO
sig Key {}
sig Time {}
sig Room {
  keys: set Key,
  currentKey: Key one -> Time }
sig Guest { gkeys: Key -> Time }
one sig FrontDesk {
  lastKey: (Room -> lone Key) -> Time,
  occupant: Room -> Guest -> Time }
fun nextKey [k: Key, ks: set Key]: set Key {
  KO/min [KO/nexts[k] & ks] }
```



# Protokol klíčů v hotelu - invarianty

```
fact {  
  all k: Key | lone keys.k  
  all r:Room, t:Time| r.currentKey.t in r.keys }  
  
pred noFrontDeskChange [t,t': Time] {  
  FrontDesk.lastKey.t = FrontDesk.lastKey.t'  
  FrontDesk.occupant.t = FrontDesk.occupant.t' }  
  
pred noRoomChangeExcept [rs: set Room, t,t': Time] {  
  all r: Room - rs | r.currentKey.t = r.currentKey.t' }  
  
pred noGuestChangeExcept [gs: set Guest, t,t': Time] {  
  all g: Guest - gs | g.gkeys.t = g.gkeys.t' }
```



# Protokol klíčů v hotelu - přihlášení

```
pred checkin [ g: Guest, r: Room, k: Key, t,t': Time ] {  
  // the guest holds the input key  
  g.gkeys.t' = g.gkeys.t + k  
  let occ = FrontDesk.occupant | {  
    // the room has no current occupant  
    no r.occ.t  
    // the guest becomes the new occupant of the room  
    occ.t' = occ.t + r->g }  
  let lk = FrontDesk.lastKey | {  
    // the input key becomes the room's current key  
    lk.t' = lk.t ++ r->k  
    // the input key is the successor of the last key in  
    // the sequence associated to the room  
    k = nextKey [r.lk.t, r.keys] }  
  noRoomChangeExcept [none, t, t']  
  noGuestChangeExcept [g, t, t'] }
```





# Protokol klíčů v hotelu - vstup do místnosti

```
pred entry [ g: Guest, r: Room, k: Key, t, t': Time ] {  
  // the key used to open the lock is one of  
  // the key the guest holding  
  k in g.gkeys.t  
  // pre and post conditions  
  let ck = r.currentKey |  
    // not a new guest  
    (k = ck.t and ck.t' = ck.t)  
    // new guest  
    or (k = nextKey [ck.t, r.keys] and ck.t' = k)  
  // frame conditions  
  noRoomChangeExcept [r, t, t']  
  noGuestChangeExcept [none, t, t']  
  noFrontDeskChange [t, t'] }
```



# Protokol klíčů v hotelu - odhlášení

```
pred checkout [ g: Guest, t,t': Time ] {  
  let occ = FrontDesk.occupant | {  
    // the guest occupies one or more rooms  
    some occ.t.g  
    // the guest's room become available  
    occ.t' = occ.t - (Room -> g)  
  }  
  // frame condition  
  FrontDesk.lastKey.t = FrontDesk.lastKey.t'  
  noRoomChangeExcept [none, t, t']  
  noGuestChangeExcept [none, t, t']  
}
```



# Protokol klíčů v hotelu - dynamika v čase

```

pred init [t: Time] {
  // no guests have keys
  no Guest.gkeys.t
  // the roster at the front desk shows no room as occupied
  no FrontDesk.occupant.t
  // the record of each room's key at the
  // front desk is synchronized with the
  // current combination of the lock itself
  all r: Room | r.(FrontDesk.lastKey.t) = r.currentKey.t }
fact Traces {
  init [T0/first]
  all t: Time - T0/last |
    let t' = T0/next [t] | some g: Guest, r: Room, k: Key |
      entry [g, r, k, t, t']
      or checkin [g, r, k, t, t'] or checkout [g, t, t'] }

```


# Protokol klíčů v hotelu - verifikace

```
assert noBadEntry {  
  all t: Time, r: Room, g: Guest, k: Key |  
    let t' = T0/next [t], o = r.(FrontDesk.occupant).t |  
      (entry [g, r, k, t, t'] and some o)  
      implies g in o  
}
```

```
check noBadEntry for 3 but 2 Room, 2 Guest, 5 Time
```



# Protokol klíčů v hotelu - protipříklad



# Protokol klíčů v hotelu - oprava

```
fact noIntervening {  
  all t: Time - T0/last |  
    let t' = T0/next [t], t'' = T0/next [t'] |  
      all g: Guest, r: Room, k: Key |  
        checkin [g, r, k, t, t']  
        implies (entry [g, r, k, t', t''] or no t'')
```

```
check noBadEntry for 3 but 2 Room, 2 Guest, 10 Time
```



# Literatura I